Fourier Optics with Maple

Herbert Bailey[*] *Azad Siahmakoun*[†]

Abstract: This paper describes an application of Maple in a first course in Fourier Optics given at Rose-Hulman Institute of Technology. The course includes a two-hour laboratory using Maple to help with computation and visualization of the important Fourier theorems. Examples are given for each of the theorems and a library of frequently used functions is included. These are then applied to solve a simple optics problem. The final section includes some "pitfalls" that we encountered and suggestions on how to avoid them.

Introduction

The primary problem in optics is to understand how optical systems process light to form images. Fourier transforms provide an elegant and practical way to describe these systems. The Fourier transforming property of a single lens enables us to generate spatial-frequency spectrum of a given input or that of its image at the back focal plane of the lens. Manipulating the spectral distribution of an input (known as spatial filtering) alters the light distribution in the output plane of an imaging system. Popular examples of spatial filtering are edge enhancement, contrast reversal, deblurring, and pattern recognition.

The first course in Fourier optics is challenging from both the mathematical and computational requirements. This note is an outgrowth of a two hour Maple lab that was introduced to help with the computations and visualization of Fourier theorems. Each student is provided with a `library' of functions that are frequently used in optics. Procedures are included to convert the Maple version of the transforms to the versions used in the course text. Currently we are using [1], the classic *Introduction To Fourier Optics* by J. W. Goodman.

There are a number of Fourier theorems that are essential in Fourier optics. Proving these theorems is a help in understanding them, but a greater help to the students is to demonstrate the theorems for specific functions. This can be done by comparing the transformed equations or, better yet, by comparing their plots. We include examples for the linearity, separability, similarity, shift and convolution theorems.

Probably the most important part of an optics course is its laboratory portion where students set up commonly used optical systems and compare the results with the theory. Some `experiments' can be done via Maple. In this note we analyze a simple optical system and show the effects of changing the pupil size of a lens on intensity distribution of the image. The final section of this note lists some pitfalls of blindly relying on the Maple transform package and suggests ways of overcoming these difficulties.

*Department of Mathematics, Rose-Hulman Institute of Technology, Terre Haute, IN 47803, Herb.Bailey@rose-hulman.edu

†Department of Physics and Applied Optics, Rose-Hulman Institute of Technology, Terre Haute, IN 47803

Functions frequently used in optics

We define functions called *pulse*, *rect*, *tri*, *gaussian*, and *sinc*.

PULSE

```
> pulse := proc(x,xL,xR)
> Heaviside(x-xL) - Heaviside(x-xR);
> end:
```

```
> plot(pulse(x,-1,2),
> x=-2..3,tickmarks=[6,3]);
```



RECT

```
> rect := proc(x)
> pulse(x,-1/2,1/2);
> end:
```

```
> plot(rect(x),x=-2..2,tickmarks=[5,3]);
```



The product of rectangular functions, rect(x/a) rect(y/b), represents in the *xy*-plane a rectangular aperature with sides of lengths *a* and *b*.

TRI

```
> tri := proc(x)
> (1+x)*pulse(x,-1,0) + (1-x)*pulse(x,0,1);
> end:
```

```
> plot(tri(x),x=-2..2,tickmarks=[5,3]);
```



GAUSSIAN

```
> gaussian := proc(x,a)
> exp(-a^2*x^2);
> end:
```

```
> plot(gaussian(x,1),
> x=-3..3,tickmarks=[7,3]);
```



SINC

```
> sinc := proc(x)
> if x = 0 then 1 else sin(Pi*x)/(Pi*x) fi;
> end:
```

```
> plot(sinc(x), x = -5..5, tickmarks=[5,3],
> scaling=constrained);
```



The sinc function is the Fourier transform of the rectangular function.

Maple and the Goodman text use different definitions for the Fourier transform and its inverse. The relationship between the definitions are given in the following table.

Fourier transform	Inverse Fourier transform
Maple	
$\int_{-\infty}^{\infty} g(x)\, e^{(-I\, w\, x)}\, dx$	$\frac{1}{2\,\pi} \int_{-\infty}^{\infty} G(w)\, e^{(I\, x\, w)}\, dw$
Goodman	
$\int_{-\infty}^{\infty} g(x)\, e^{(-2\, I\, \pi\, fX\, x)}\, dx$	$\int_{-\infty}^{\infty} G(x)\, e^{(2\, I\, \pi\, x\, fX)}\, dfX$

To make it easier for the student to use Maple in support of learning from the Goodman text we wrote the procedures Fourier and invFourier which implement in Maple the Goodman definitions of the transform and its inverse. Optical systems are usually two dimensional, and Goodman uses fX and fY to represent spatial frequencies along the *x*-axis and *y*-axis, respectively. The Maple command for the one dimensional Fourier transform is fourier(g,x,w). For the Goodman version we use Fourier(g,x, fX).

FOURIER

```
> Fourier := proc(g,x,fX)
> local w,v;
> v := inttrans[fourier](g,x,w);
> evalc(subs(w=2*Pi*fX,v));
> end:
```

INVFOURIER

```
> invFourier := proc(G,fX,x)
> local v;
> v := inttrans[invfourier](G,fX,x);
> evalc(subs(x=x*2*Pi,2*Pi*v));
> end:
```

Fourier theorems

In this section we state some of the Fourier theorems that are used in Fourier optics computations. To assist in understanding these theorems, we apply them to specific functions and, in some cases, show the results graphically.

SEPARABLE FUNCTIONS

A function of two variables f(x,y) is separable if f(x,y) = g(x) h(y). In this case the Fourier transform of f(x,y) is equal to the product of the transforms of g(x) and h(y), that is

$$\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} \mathrm{f}(x, y)\, e^{(-2\pi I(x\,fX+y\,fY))}\,dx\,dy = \int_{-\infty}^{\infty} \mathrm{g}(x)\, e^{(-2\pi I x\,fX)}\,dx \int_{-\infty}^{\infty} \mathrm{h}(y)\, e^{(-2\pi I y\,fY)}\,dy$$

We demonstrate this theorem by choosing the separable function f(x,y) = g(x) h(y) = rect(x/a) rect(y/b). This function represents a rectangular slit with width *a* and height *b*. Since the Maple transform package does not include two dimensional transforms, we just integrate twice to transform f(x,y). Also, since rect is an even function with finite domain, we can obtain the Fourier transform of f(x,y) with the corresponding cosine transform. We show the calculations below for the case *a* = 1 and *b* = 5. Note that in this case $f(x, y) = 0$ for $|x| > 1/2$ and $|y| > 5/2$.

```
> F:=
> int(int(rect(x)*rect(y/5)*
>      cos(-2*Pi*(x*fX+y*fY)),
> x=-1/2..1/2),y=-5/2..5/2);
```

$$F := 2\,\frac{\cos\left(\frac{5}{2}\pi\,fY\right)\sin\left(\frac{5}{2}\pi\,fY\right)\sin(\pi\,fX)}{\pi^2\,fY\,fX}$$

```
> plot3d(F,fX=-2..2,fY=-1..1,axes=FRAMED,
> tickmarks=[5,5,3]);
```




```
> G:=Fourier(rect(x),x,fX);
```

$$G := \frac{\sin(\pi\,fX)}{\pi\,fX} + 2\,I\sin(\pi\,fX)\,\pi\,\mathrm{Dirac}(2\,\pi\,fX)$$

```
> H:=Fourier(rect(y/5),y,fY);
```

$$H := \frac{\sin(5\,\pi\,fY)}{\pi\,fY} + 2\,I\sin(5\,\pi\,fY)\,\pi\,\mathrm{Dirac}(2\,\pi\,fY)$$

```
> plot3d(G*H,fX=-2..2,fY=-1..1,axes=FRAMED,
> tickmarks=[5,5,3]);
```



The plots look the same and with a little computation we see that F = GH, thus the Separability theorem is demonstrated for this example.

SHIFT THEOREM

If $\mathrm{f}(x, y)$ = invFourier{ $\mathrm{F}(fX, fY)$ }, then
$\mathrm{f}(x-a, y-b)$
= invFourier{ $\mathrm{F}(fX, fY)\, e^{(-I\,2\,\pi\,(a\,fX+b\,fY))}$ }.

To demonstrate this theorem, we let $a = 5$, $b = 2$, $\mathrm{f}(x, y) = \mathrm{g}(x)\,\mathrm{h}(y) = e^{(-4\,x^2)}\,e^{(-y^2)}$. We first find the Fourier transform of f(x,y) using separabilty. We then take the inverse transform of $\mathrm{F}(fX, fY)\, e^{(-I\,2\,\pi\,(a\,fX+b\,fY))}$ to obtain $\mathrm{f}(x-a, y-b)$.

```
> g:=gaussian(x,2); h:=gaussian(y,1);
```

$$g := e^{(-4\,x^2)}$$

$$h := e^{(-y^2)}$$

```
> G:=Fourier(g,x,fX); H:=Fourier(h,y,fY);
```

$$G := \frac{1}{4}\sqrt{4}\sqrt{\pi}\,e^{(-1/4\,\pi^2\,fX^2)}$$

$$H := \sqrt{\pi}\,e^{(-\pi^2\,fY^2)}$$

Then by the separability theorem we have F(fX,fY) = G (fX) H(fY). We again apply separability to calculate invFourier{ $\mathrm{F}(fX, fY)\, e^{(-I\,2\,\pi\,(a\,fX+b\,fY))}$ }. The terms involving x are denoted by g1 and those involving y are denoted by h1.

```
> g1:= invFourier(G*exp(-I*2*Pi*5*fX),fX,x);
```

$$g1 := \sqrt{\pi}\sqrt{\frac{1}{\pi}\,e^{\left(-\frac{(2\,\pi\,x-10\,\pi)^2}{\pi^2}\right)}}$$

```
> h1:=invFourier(H*exp(-I*2*Pi*2*fY),fY,y);
```

$$h1 := \sqrt{\pi}\sqrt{\frac{1}{\pi}}\, e^{\left(-1/4\,\frac{(2\,\pi\,y-4\,\pi)^2}{\pi^2}\right)}$$

```
> plot3d(\{g1*h1,g*h\},x=-2..8,y=-2..8,
> axes=FRAMED);
```



The gaussian peak in the background is the original function. The peak in the foreground shows the shift of 5 units in the x direction and two units in the y direction.

SIMILARITY THEOREM

If Fourier{ $g(x, y)$ } = $G(f_X,\ f_Y)$, then

Fourier $\{g(a\,x,\ b\,y)\} = \frac{1}{|a\,b|}\ G\left(\frac{f_X}{a},\ \frac{f_Y}{b}\right)$.

We now demonstrate the Similarity theorem, in one dimension, with $g(x) = \text{sinc}(x)^2$ and with $a = 3$.

```
> g:=x->sinc(x)^2:
```

```
> plot(g(x),x=-2..2,tickmarks=[7,3]);
```




```
> plot(Fourier(g(x),x,fX),fX=-4..4,
> tickmarks=[9,3]);
```




```
> plot(g(3*x),x=-2..2,tickmarks=[7,3]);
```




```
> plot(Fourier(g(3*x),x,fX),fX=-4..4,
> tickmarks=[9,3]);
```



In this example we see that the Fourier transform of the sinc-squared function is the triangle function. The transform is flattened and widened when x is replaced by $3\,x$, as predicted by the Similarity theorem. The peak value of $g(x)$

itself is not changed, but its graph is sharper when x is replaced by $3\,x$. The Similarity theorem shows that contraction of functions in the space domain leads to the broadening of their Fourier transforms in the spatial frequency domain, provided a is greater than one. The relations are reversed for a between zero and one.

CONVOLUTION THEOREM

If Fourier $\{g(x,\,y)\} = G(fX,\,fY)$ and Fourier $\{h(x,\,y)\} = H(fX,\,fY)$, then

$$\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} g(s,\,t)\,h(x-s,\,y-t)\,ds\,dt = \text{invFourier}\{G(fX,\,fY)\,H(fX,\,fY)\},$$

where the double integral is the convolution of g and h.

We now demonstrate the theorem for the one dimensional case with g(x) = Dirac(x+5) + Dirac(x-5) and h(x) = sinc(x).

```
> g:=x->Dirac(x+5)+Dirac(x-5):
> h:=x->sinc(x):
> G:=Fourier(g(x),x,fX);
```

$$G := 2\cos(10\,\pi\,fX)$$

```
> H:=Fourier(h(x),x,fX);
```

$$H := (\frac{1}{2}\pi\,(\text{Heaviside}(-2\,\pi\,fX+\pi) - \text{Heaviside}(2\,\pi\,fX-\pi)) - \frac{1}{2}\pi\,(\text{Heaviside}(-2\,\pi\,fX-\pi) - \text{Heaviside}(2\,\pi\,fX+\pi)))/\pi$$

```
> convo:=int(g(s)*h(x-s),s=-6..6);
```

$$convo := \frac{\sin(\pi\,(x+5))}{\pi\,(x+5)} + \frac{\sin(\pi\,(x-5))}{\pi\,(x-5)}$$

Note that $g(s) = 0$ for $|s| > 6$.

```
> plot(convo,x=-10..10);
```




```
> FT_convo:=Fourier(convo,x,fX):
> GH:=G*H:
> plot(FT_convo,fX=-1..1);
```




```
> plot(GH,fX=-1..1);
```



Thus for this example we see that the Fourier transform of the convolution of two functions is equal to the product of their Fourier transforms and the theorem is demonstrated.

An optics problem



Plane waves of monochromatic light (with unit amplitude and wave length = .0001 cm) illuminate a square aperture (sides = 0.05 cm) located at the object plane as shown in the above Figure. The transmitted light travels through free space to a square pupil (sides = 1 cm) and a converging lens of 10 cm focal length. The light transmitted by the lens propagates through free space to the image plane. We are to find the image intensity distribution, I_i, in the image plane. Equations describing this system are derived in Chapter V of Goodman. The vertical axis in the figure is the x-axis. The y-axis is perpendicular to the plane of the paper and is not shown. The intensity distribution in one dimension is expressed as the square of a convolution,

$$I_i(x) = (ug \, \mathrm{O} \, h)^2 = \left(\int_{-\infty}^{\infty} \mathrm{ug}(s)\, \mathrm{h}(x-s)\, ds\right)^2,$$

where ug is related to the square of aperture function in the object plane, h is related to the input response of the lens and pupil to a point input (impulse response), and the symbol O is used to denote the convolution. For our problem (in one dimension), ug and h are given by:

$$ug = \mathrm{rect}(20x) \text{ and } h = \mathrm{invFourier}\{\mathrm{rect}(\mathrm{fX}/500)\}.$$

We now compute the inverse transform and convolution and plot the intensity distribution.

```
> ug:= rect(20*s);
```

$$ug := \mathrm{Heaviside}(20\,s + \frac{1}{2}) - \mathrm{Heaviside}(20\,s - \frac{1}{2})$$

```
> h:=invFourier(rect((fX/500)),fX,x);
```

$$h := \frac{\sin(500\,\pi\,x)}{\pi\,x} - 2\,I\,\pi\,\mathrm{Dirac}(2\,\pi\,x)\sin(500\,\pi\,x)$$

```
> h1:=subs(x=x-s,h);
```

$$h1 := \frac{\sin(500\,\pi\,(x-s))}{\pi\,(x-s)} - I\,\mathrm{Dirac}(x-s)\sin(500\,\pi\,(x-s))$$

```
> Fi:=int(ug*h1,s=-1/40..1/40);
```

$$Fi := \frac{-\mathrm{Si}(-\frac{25}{2}\,\pi + 500\,\pi\,x) + \mathrm{Si}(\frac{25}{2}\,\pi + 500\,\pi\,x)}{\pi}$$

We can use these finite limits since $ug(s) = 0$ for $|s| > 1/40$.

```
> Ii:=(Fi)^2:
> plot(Ii,x=-0.03..0.03,tickmarks=[7,3]);
```



The above calculations were restricted to the one dimensional case. For this optical system (and for many others) the functions are separable and thus Ii(x, y) = Ii(x)Ii(y). A plot of Ii(x, y) is shown below.

```
> plot3d(Ii*subs(x=y,Ii),x=-0.03..0.03,
> y=-0.03..0.03, tickmarks=[7,7,3],
> axes=FRAMED);
```



If the sides of the pupil are halved then the function h is changed. We calculate the resulting image intensity below, where a trailing s is used to denote the results with the smaller aperture.

```
> hs:=invFourier(rect(fX/250),fX,x):
> h1s:=subs(x=x-s,hs):
> Fis:=int(ug*h1s,s=-1/40..1/40):
> Iis:=(Fis)^2:
> plot(Iis,x=-0.03..0.03,tickmarks=[7,3]);
```



Notice that the intensity modulation of the image (fringes) is due to coherent illumination of the object and finite size of the square pupil (diffraction limited imaging).

The accuracy of the Fourier technique in optical processing is astonishing. It makes a powerful teaching and research tool when its recipes are prescribed with the ease of Maple .

Some pitfalls

We tried to define the rectangular function rect(x) via the piecewise function rather than the Heaviside function. This failed, as shown below, since the Maple Fourier package would not transform the piecewise function.

```
> rectp:=x->piecewise(x<-1/2,0,x<1/2,1,0):
> R:=inttrans[fourier](rectp(x),x,w);
```

$$R := \mathrm{fourier}\left(\begin{cases} 0 & x < \frac{-1}{2} \\ 1 & x < \frac{1}{2} \\ 0 & otherwise \end{cases}, x, w\right)$$

In one application of the Convolution Theorem , we tried to apply **invFourier** to the product of a *sinc* function and a *rect* function. The Maple package was "unable to handle". This difficulty was overcome by noting that $e^{(2\,I\,\pi\,x\,fX)} = \cos(2\,\pi\,x\,fX) + I\sin(2\,\pi\,x\,fX)$ and applying the Fourier inversion integral. For an even function, H(fX) = H(-fX), so the only contribution to the inversion integral will come from the cosine term. Also, if the function vanishes outside a certain domain then we need only integrate over the domain where the function is nonzero. These ideas are shown in the following calculations.

```
> F:=2*sinc(2*fX):
> G:=rect(fX):
> q:=invFourier(F*G,fX,x);

Error, (in evalc/int)
Unable to handle definite integral
```

Since F and G are both even functions and G = 0 for $|fX| > 1/2$, the inverse transform is given by:

```
> q1:=
> simplify(int(F*G*cos(2*Pi*x*fX),
> fX=-1/2..1/2));
```

$$q1 := -\frac{\mathrm{Si}(\pi\,(-1+x)) - \mathrm{Si}(\pi\,(x+1))}{\pi}$$

This result can be verified by the Convolution theorem after noting that the integrand in the convolution integral has compact support.

```
> ff:=invFourier(F,fX,x):
> gg:=invFourier(G,fX,x):
> QQ:=
> int(subs(x=s,ff)*subs(x=s-x,gg),s=-5..5);
```

$$QQ := \frac{\mathrm{Si}(\pi - \pi\,x) - \mathrm{Si}(-\pi\,x - \pi)}{\pi}$$

Agreement is obtained by noting that Si(x) is an odd function.

The final pitfall that we have noted involves integration of the exponential form of the sinc function. Maple gives an incorrect integral. This difficulty can be corrected by applying the **evalc** command to the exponential form before integrating. Note that in our library of transform procedures, we have included the **evalc** command. This pitfall and cure are demonstrated in the following calculations.

```
> h:=inttrans[fourier](rect(x),x,w);
```

$$h := e^{(1/2\,I\,w)}\,(\pi\,\mathrm{Dirac}(w) - \frac{I}{w}) - e^{(-1/2\,I\,w)}\,(\pi\,\mathrm{Dirac}(w) - \frac{I}{w})$$

```
> q:=int(h,w);
```

$$q := 2\,\mathrm{Si}(\frac{1}{2}\,w) - \pi\,\mathrm{csgn}(w)$$

The correct expression for this integral is $2\,\mathrm{Si}(\frac{w}{2})$. Maple gives the correct answer if h is separated into its real and imaginary parts before integration. This is done below,

```
> he:=evalc(h);
```

$$he := 2\,\frac{\sin(\frac{1}{2}\,w)}{w} + 2\,I\sin(\frac{1}{2}\,w)\,\pi\,\mathrm{Dirac}(w)$$

```
> q1:=int(he,w);
```

$$q1 := 2\,\mathrm{Si}(\frac{1}{2}\,w)$$

Remarks

Note that the Dirac functions in the above expressions can be ignored since Dirac(w) only contributes when w = 0, and when w = 0 the Dirac functions are multiplied by zero. Thus h = he = sin(w/2)/(w/2) and q1 is the correct integral.

Summary

These difficulties that have been encountered can be avoided by: (1) not using the piecewise function, (2) using the cosine form of the Fourier transform for transforming even functions, (3) using finite limits when the domain of the functions are finite, and (4) applying the **evalc** command to transforms before making additional computations.

Acknowledgment

Our thanks to Robert Lopez for his valuable suggestions during the preparation of this manuscript.

References

[1] Joseph W. Goodman: *Introduction to Fourier optics*, Second Edition, McGraw-Hill, (1996).

Biographies

Herbert Bailey did his undergraduate work at Rose-Hulman Institute of Technology in Chemical and Electrical Engineering. His Ph.D. in Mathematics was from Purdue University. He spent fifteen years as a mathematician in industry and thirty years teaching mathematics. He retired from Rose-Hulman in 1992. His current interests include astronomy, optics and geometry.

Azad Siahmakoun received his B.S. degree in Physics from Joundi Shahpoor University in 1978, his M.S. degree in Physics from Texas A&I University in 1981, and his Ph.D. from University of Arkansas in 1988. For his doctoral dissertation he investigated the temporal and spatial properties of a phase-conjugate laser. He joined Rose-Hulman Institute of Technology in September of 1987 where he is presently an associate professor of physics and applied optics. His teaching interests are laser physics, electro-optics, microsensors, nonlinear optics, and Fourier optics applications. Dr. Siahmakoun's interests in research include: Optical Phase-Conjugation, Holographic Interferometry, Nonlinear Dynamics and Chaotic systems, Applications of Photorefractive crystals & polymer thin films in image storage and processing.

Fitting Logistics to the U. S. Population

*William C. Bauldry**

Abstract: We consider fitting a logistic model to the U. S. population in the light of modern courses in differential equations. A direct logistic fit to the data with least squares is shown to be intractable. An approach to fitting using divided differences is given and a novel model is developed. We close with suggested directions for student explorations.

Historical background

> Prediction is difficult, especially when it involves the future. — Mark Twain

Modeling human populations dates back to Graunt's 1662 treatise, *Natural and Political Observations* [1], on the mortality tables of London; many consider him to be the founder of demographic analysis. Graunt realised that his estimate of 64 years for the doubling time of London's population, when applied to the world in general, "shall produce far more People, than are now in it." (The political economist, Petty, proposed a declining growth rate in 1683.) The most influential text is British cleric Rev. Malthus' 1798 apocalyptic work, *An Essay on the Principle of Population* [2]. Malthus predicted the world population would grow exponentially with disastrous results. This essay is still widely read today. (Appalachian's library has eleven copies of various editions.) The next advance is due to the Belgian Verhülst who introduced the *logistic* model in his 1845 memoir [3]. Verhülst modified the constant growth rate of the exponential model so as to depend on population size. Limits in demographic data collection and calculational ability prevented Verhülst from making quantitative predictions, but he could show a limiting maximum population. Americans Pearl and Reed (with several sets of coauthors) wrote a series of very controversial papers (see [4] for an interesting discussion of the controversy) from 1920 to 1940 modeling the U. S. population logistically. Pearl was convinced the logistic was the "universal population model," and spent many years on further refinements and extensions of the logistic equation. Pearl and Reed were originally unaware of Verhülst's work.

Recent work gives many reasons why most models, logistics in particular, are unsuitable for predicting human population levels. For instance, current models can not account for increases in technology that affect carrying capacity or social upheaval. (See, e.g., Marchetti, Meyer, and Ausubel [5].) Nevertheless, there is value in a logistic fit: Collected data can be tested for reasonableness. Estimates made between one census and the next can be checked. Short term predictions can be made. Changes and trends in growth rate can be sensed. Et cetera.

*Appalachian State U, Boone, NC, wmcb@math.appstate.edu

U. S. population counts

Yearly population values are available from the Historical Data section of the Census Bureau's web pages (www.census.gov). We collected the counts and defined `Census` as a Maple list of pairs `[year,count]`. Enter the decile U. S. census counts.

```
> Census := [     [1790,3929],   [1800,5297],
>   [1810,7224],   [1820,9618],   [1830,12901],
>   [1840,17120],  [1850,23261],  [1860,31513],
>   [1870,39905],  [1880,50262],  [1890,63056],
>   [1900,76094],  [1910,92407],  [1920,105683],
>   [1930,123188], [1940,132122], [1950,151684],
>   [1960,180671], [1970,204875], [1980,227220],
>   [1990,249924]]:
> N := nops(Census);
```

The population values shown are in millions.

For student explorations, the file `YearlyCensusData.m` containing the yearly numbers from 1670 to 1991 is available from

www.mathsci.appstate.edu/~wmcb/USPop/

To read the data into the variable `Pop`, execute

```
> read `YearlyCensusData.m`:
```

Now, on to exploring how to fit a logistic model.

Logistic models and data fitting

The Malthusian model is based on the principle

$$\textit{growth} \propto \textit{population size}.$$

Expressed as a differential equation, this model is

$$\frac{dP}{dt} = r \cdot P$$

where r is the constant of proportionality, the growth rate. It only takes a moment to realize the shortcomings of this model. Verhülst's extension was to replace the constant growth rate with a decreasing rate proportional to the density in terms of the maximum possible population.

$$\frac{dP}{dt} = r \cdot \left(1 - \frac{P}{P_{max}}\right) \cdot P$$

He called this model *logistic* (though we don't quite know why). The logistic differential equation is easy to solve by separating variables and then integrating via partial fractions — "homework lite." We find that

$$P(t) = \frac{P_{max}}{1 + \left(\frac{P_{max}}{P_0} - 1\right) e^{-rt}}$$

Define the function:

```
> logistic :=
>   (r,M,Po,t) -> M/(1+(M/Po-1)*exp(-r*t));
```

where we've used M for P_{max} and Po for P_0.

We cannot directly use Maple's leastsquare function from the fit subpackage of the stat package, since the logistic equation is not linear in the parameters r, P_{max}, and P_0. Apparently, we'll have to do the fit manually.

A DIRECT APPROACH

Define the sum of the squares of the errors as SSE.

```
> SSE := sum(
>   ( Census[i][2]
>     - logistic(r,M,Po,Census[i][1]) )^2,
>   i=1..N):
```

Use standard calculus techniques to minimize SSE.

```
> params := {r, M, Po}:
> eqns := map2(diff, SSE, params):
```

Our first step is to find critical points by finding the zeros of the system eqns in terms of the values of params. While theoretically trivial, finding the roots turns out to be very difficult since this system of equations is highly nonlinear.

```
> _time := time():
> fsolve(eqns, params);
> (time() - _time) * `seconds`;
```

```
Error, (in fsolve/genroot) cannot converge
to a solution
```

$$168.000\ seconds$$

Even supplying fsolve with search ranges (based on hindsight) doesn't work. Try:

```
> _time := time():
> fsolve(eqns, params,
>   {r=0.02..0.03, M=289000..290000,
>    Po=3900..4000});
> (time() - _time) * `seconds`;
```

Using Ross Taylor's Newton procedure from the *share library* (see ?share) doesn't help. However, Newton does give valuable clues as to what's going wrong. To read the procedure from the library and set up its use, enter:

```
> with(linalg):
> with(share);
> readshare(Newton, numerics);
> NewtonEqns := convert(eqns, list):
```

Newton is a procedure in the share library numerics. We had to convert the system of equations from a set into a list as required by Newton. We also needed the linalg package since Newton's calculations are vector based. Now give it a try using $r = 0.03$, $P_{max} = 289000$, and $P_0 = 3900$.

```
> _time := time():
> Newton(NewtonEqns,
>   [r=0.03, M=289000, Po=3900],
>   output={norm,variables},steps=teststep);
> (time() - _time) * `seconds`;
```

(See ?Newton for an explanation of the options output and steps used above.) The output shows the parameters given more and more infeasible values. Using gradients in a steepest-descent method also fails. (A Maple worksheet implementing steepest descent is also available at

www.mathsci.appstate.edu/~wmcb/USPop/

To help understand why the problem is intractable, consider the following. Reduce SSE to two variables by setting $P_0 = 3929$ and look at a contour plot.

```
> SSE2 := subs(Po=3929, SSE):
> with(plots):
> contourplot(SSE2,
>   r=0.01..0.04, M=288000..290000,
>   grid=[35,35], filled=true);
```



The change in the surface with respect to r is extremely small compared to that with respect to M. It's quite difficult to find a minimum on so uniform a surface. This problem is similar to "stiffness" encountered in differential equations.

USING DIVIDED DIFFERENCES

We've quickly found that a direct attack using least squares to estimate the parameters is fruitless. The expressions are intractable. Using the dynamics of the model to estimate the parameters r, P_{max}, and P_0 makes for much more interesting analysis and gives students a much better appreciation of differential equations and modeling in practice.

We are lead to another method of estimating the parameters by considering the relative growth rate P'/P.

$$\begin{aligned} \frac{dP}{dt}\cdot\frac{1}{P} &= r\cdot\left(1-\frac{P}{P_{max}}\right) \\ &= a+bP \end{aligned}$$

where $a = r$ and $b = -r/P_{max}$. Discretizing this formula with a divided difference gives us

$$\frac{\Delta_1 P}{P} = a + bP.$$

We have an expression linear in $\Delta_1 P/P$ and P where the coefficients a and b will yield r and P_{max}.

We still need P_0. There is no reason to prefer any one data point over another as an initial condition. In the logistic equation, substitute $P := 1/N$ and $T := exp(-rt)$ to yield

$$\begin{aligned} N &= \frac{1}{P_{max}} + \left(\frac{1}{P_0} - \frac{1}{P_{max}}\right)\cdot T \\ &= c + dT. \end{aligned}$$

where $c = 1/P_{max}$ and $d = (1/P_0 - 1/P_{max})$. We have a second linear equation that can be used to determine P_0. Since the desired parameters appear linearly in these equations, we can now use Maple's `leastsquare` for the calculations.

First, load needed packages and define a set of utilities.

```
> with(stats): with(fit):
> alias(
>   LS=leastsquare[[x,y], y=a+b*x, {a,b}]):
> shift := (L,s) ->
>   map(unapply([x[1]-s, x[2]] ,x), L);
```

The outline of the fitting procedure is:

1. Shift the data so the abscissas begin at 0.
2. Calculate the relative rates using the relative symmetric divided difference

$$\frac{\Delta_1 P_i}{P_i} := \frac{P_{i+1} - P_{i-1}}{t_{i+1} - t_{i-1}} \cdot \frac{1}{P_i}.$$

3. Use least squares on $a + bP$ to estimate r and P_{max}.
4. Apply $exp(-rt)$ to each year t.
5. Calculate the reciprocals of the population counts.
6. Use least squares on $c + dT$ to estimate P_0.

Enter the fitting function.

```
> LogisticFit := proc(L)
> local S, P, RelDivSymmDiff, i, LS_fit,
>   r, Pmax, R, X, Po;
>  S := shift(L,L[1,1]);
>  P := S[2..nops(S)-1,2];
>  RelDivSymmDiff := [seq(
>    (S[i+1,2]-S[i-1,2])/(S[i+1,1]-S[i-1,1])
>    /S[i,2], i=2..nops(S)-1)];
>  LS_fit := rhs(LS([P,RelDivSymmDiff]));
>  r := coeff(LS_fit,x,0);
>  Pmax := -r/coeff(LS_fit,x,1);
>  X := evalf(map(unapply(exp(-r*x),x),
>    S[1..nops(S),1]));
>  R := map(x->1/x[2], L);
>  LS_fit := rhs(LS([X,R]));
>  Po := 1/
>    (coeff(LS_fit,x,0)+coeff(LS_fit,x,1));
> RETURN(r,Pmax,Po);
> end;
```

Now estimate the parameters for the data `Census`.

```
> evalf(LogisticFit(Census));
```

$$.02916900417, 289412.0022, 3987.273155$$

Note how quickly Maple calculated the parameters. We put these values into `logistic` to have the desired logistic function. Remember to translate with $t - 1790$ to account for the shift done in `LogisticFit`.

```
> the_fit := logistic(",t-1790):
> fnormal(the_fit, 6);
```

$$the\_fit := \frac{289412.}{1 + 71.5839\, e^{(-.0291690\, t + 52.2125)}}$$

Check `the_fit` with a plot.

```
> display([ plot(the_fit, t=1790..2000),
>   plot(Census,style=point,symbol=diamond)],
>   view=[1790..2000, 0..270000]);
```



Reed, Pearl, and Kish's 1940 logistic fit to the U. S. population (from 1790 to 1940) had $P_{max} = 184$ (million) and $r = 0.0322$ [6]. (See §4.5 of Yeargers, Shonkwiler, and Herrod [7] for a different method of transforming the logistic equation to be linear in the parameters.)

POINTS FOR CLASS DISCUSSION

1. What are the parameter estimates for the full data set?
2. How do the above estimates compare to Reed, Pearl, and Kish's fit?
3. Which of symmetric, forward, or backward divided differences gives the best fit?

An excursion into an alternate model

It is clear from the plot above that there are problems with the logistic fit beginning around 1940, a time of great societal upheaval and technological advancement. On the other hand, Pearl, Reed, and Kish's fit is quite reasonable up to 1940. We can posit that the changes in society about 1940 caused a change in growth rate (large immigration) and carrying capacity (rapid industrialization). To account for this change, we fit two separate logistics, allowing an artificial discontinuity at 1940 as follows:

```
> Stage1 := Census[1..16]:
> Stage2 := Census[16..N]:
> first := logistic(
>   evalf(LogisticFit(Stage1)),t-1790):
> second := logistic(
>   evalf(LogisticFit(Stage2)),t-1940):
> the_fit := t ->
>   piecewise(t<1940, first, second):
> fnormal(the_fit(t), 6);
```

$$\begin{cases} \dfrac{187186.}{1+47.0151\, e^{(-.0319240\, t+57.1440)}} & t < 1940 \\ \dfrac{342869.}{1+1.62291\, e^{(-.0293698\, t+56.9774)}} & otherwise \end{cases}$$

Note the dramatic shift in both growth rate and maximum population values between the two stages. Check `the_fit` with a plot.

```
> display([
>   plot(the_fit(t), t=1790..2000,
>     discont=true),
>   plot(Census, style=point,
>     symbol=diamond)],
>   view=[1790..2000, 0..270000]);
```



For an extension of these concepts, see [5] for an example of a piecewise fit of four logistics corresponding to four major economic periods in U. S. history. In addition, Maple worksheets defining other models are available at the web site containing the full data set.

Student projects

There are many sources of data on the world wide web that students may use. There are also simple experiments, requiring minimal equipment, that groups may perform. Possible directions for student projects that involve fitting logistics include:

- U. S. Population
  1. Fit a logistic, using the full data set, to the population from 1940 onward. How does the new fit compare to the one developed above? What does the new fit predict the current population to be? Compare with the U. S. Census Bureau's value from the *Population Clock* web page at

     www.census.gov/cgi-bin/popclock

  2. Investigate the effects adding Alaska and Hawaii had on the population counts and fits.
- Experimental Populations
  1. Perform R. L. Pearl's 1920 experiment with *Drosophila M.* See, e.g., Pearl's [8].
  2. Perform Carlson's mass experiment with *Brewer's yeast.*
- Diffusion
  1. Model the "adoption" of a favorite group's new musical release.
  2. Model the adoption of a new technology; e.g., Intel's "MMX" processors.

For a general survey on modeling diffusion, see [9].

Undergraduate research project[1]

Many modifications have been made to the logistic model to improve the fit. One of these is to replace the *carrying capacity*, or P_{max}, with a function of t. Arguments can be made for constant, linear, or even logistic functions for $P_{max}(t)$. As technological improvements are adopted, resource utilization is improved allowing more individuals to be carried. However, natural restrictions on resource availability limit

[1]The germ of this project was J. Herrod's manuscript "The Changing Carrying Capacity as Predicted by the U. S. Population Data."

the ability to increase the carrying capacity. Hence, a logistic carrying capacity is feasible.

Assume the *carrying capacity* is logistic. Insert a logistic term for P_{max} in the logistic differential equation. Analyze the resulting dynamics. Create a procedure for estimating the parameters for this *logistic-logistic* model.

References

[1] J. Graunt: *Natural and political observations mentioned in a following index, and made upon the bills of mortality*, Roycroft, London, (1662).

[2] Rev. T. R. Malthus: *An essay on the principle of population, or, A view of its past and present effects on human happiness: with an inquiry into our prospects respecting the future removal or mitigation of the evils which it occasions*, Johnson, London, (1798), and Cambridge U Press, England, (1992).

[3] P.-F. Verhülst: "Recherches mathématique sur la loi d'accroissement de la population," Nouveaux Mémoirs de l'Academie Royale des Sciences et Belles-Lettres de Bruxelles 18, pp.3–38, (1845).

[4] S. Kingsland: *Modeling nature*, U of Chicago Press, Chicago, (1985).

[5] C. Marchetti, P. Meyer, and J. Ausubel: "Human population dynamics revisited with the logistic model: how much can be modeled and predicted," *Tech. Forecasting and Social Change* **52**, pp. 1–30, (1996).

[6] R. Pearl, L. Reed, and L. Kish: "The logistic curve and the census count of 1940," Science **92**, pp. 486–488, (1940).

[7] E. Yeargers, R. Shonkwiler, and J. Herrod: *An introduction to the mathematics of biology*, Birkhäuser, Boston, (1996).

[8] R. Pearl: *The biology of population growth*, Knopf, New York, (1925).

[9] N. Nawaz-Sharif and K. Ramanathan: "Temporal models of innovation diffusion," IEEE Trans. on Engineering Management, EM–31, **2**, pp. 76–86, (1984).

Biography

Bill Bauldry received a Ph.D. in Approximation Theory from Ohio State University in a non-census year; however, he did appear in one of the decile counts this century. As a professor of mathematics at Appalachian State University, he combines his interests in Maple, technology, and CAS in the classroom. Maple is still his favorite syrup.

Guidelines for Authors

Any user of Maple who has implemented an original, useful, elegant, or illustrative solution to a particular problem is invited to submit a description of the solution to *MapleTech*. We especially encourage articles which have both technical merit and are accessible to the general user of Maple. Three categories of articles are published in *MapleTech*:

(i) Articles on Maple functionality.

(ii) Articles about the use of Maple in education.

(iii) Articles on research applications of Maple.

Articles in category (i) are usually provided by the Maple developers. Articles of category (ii) must be as accessible as possible. Applications in *all* areas (economics, engineering, environmental studies, etc.) are accepted. The intent and category of the article must be made clear by authors within their articles.

All articles should be roughly 4 to 8 pages long, as they would appear in *double-column* format, a 10pt font, on 8.5 x 11 inch paper. The article must be in the form of *Unix compatible* LaTeX file *in* MTN *format*. It is important that any results obtained by Maple be *reproducible* by the general user. Within the article itself, Maple input must have the form:

```
\begin{mapleinput}
int( exp( -x^2 ), x = 0..infinity );
\end{mapleinput}
```

The corresponding Maple output should be a LaTeX equation of the form:

```
\begin{maplelatex}
\[
{\frac {\sqrt {\pi }}{2}}
\]
\end{maplelatex}
```

To further ensure reproducibility of results, the author must use a standard and current version of Maple and not his or her own customized version. All articles developed in Maple worksheets should can be converted into LaTeX using the "Export to LaTeX" facility available as of Maple V Release 3. References, citations etc... should be included in MTN format.

Guidelines and tools for preparing articles for the Maple Technical Newsletter, as well as LaTeX style files and sample articles can be obtained via anonymous ftp at ftp.cecm.sfu.ca under the directory pub/MapleTech at Simon Fraser University. *Articles must be written according to the detailed instructions contained within the* **README** *file* (this file explains the MTN format). Each article must include an abstract, keywords and a short one-paragraph biography from each of the author(s) which includes fields of interest. Upon submission, the author(s) must email a postscript file containing their entire article (including figures and plots) to the editors. The postscript file will then be sent for review. The LaTeX source files plus postscript files for each of the figures are to be submitted *only* to the assistant-editor *and only after* the article has been refereed, finalized and accepted by the appropriate section editor.

All submissions are done via electronic mail — **HARD COPY WILL NOT BE ACCEPTED**. Section editors **WILL NOT** process LaTeX source files. If the article is accepted, authors will be sent a copyright form to sign and an order form for reprints. For extra reprints, please contact Birkhäuser, Boston (address and phone numbers supplied by the front cover information).

Any inquiries or submission should be directed (via email) to the Editor-in-chief Tony Scott at:

INRIA-Rocquencourt
B.P. 105
78153 Le Chesnay Cedex
France
FAX = 01 39 63 57 86
Phone = 01 39 63 51 01
e-mail: Tony.Scott@inria.fr
OR tcscott@maths.oxford.ac.uk
OR scott@cfa.harvard.edu
OR Tony.Scott@inria.fr
OR tcscott@bgumail.bgu.ac.il